```1c
    ДокументОбъект = Документы.ОперацияБух.СоздатьДокумент();
    ДокументОбъект.Дата = ДатаПоступлениеНаРасчетныйСчет - 1;
    ДокументОбъект.Организация = Данные.ОрганизацияМорозовИП;
    ДокументОбъект.СуммаОперации = Данные.СуммаДокумента;
    ДокументОбъект.Ответственный = ПараметрыСеанса.ТекущийПользователь;
    ДокументОбъект.Содержание = "УК "Равновесие" " + Реализация.Контрагент.Наименование;
    ДокументОбъект.Комментарий = "операция создана обработкой";
    ДокументОбъект.Проведен = Истина;
    Попытка
    ДокументОбъект.Записать(РежимЗаписиДокумента.Запись);
    Данные.Реализация = ДокументОбъект.Ссылка;
    Исключение
    Отказ = Истина;
    Сообщить(ОписаниеОшибки());
    КонецПопытки;
    Если Не Отказ Тогда
    УстановитьПривилегированныйРежим(Истина);
    НаборЗаписей = РегистрыБухгалтерии.Хозрасчетный.СоздатьНаборЗаписей();
    НаборЗаписей.Отбор.Регистратор.Установить(ДокументОбъект.Ссылка);
    НаборЗаписей.Прочитать();
    НоваяЗапись = НаборЗаписей.Добавить();
    НоваяЗапись.Период = ДокументОбъект.Дата;
    НоваяЗапись.Регистратор = ДокументОбъект.Ссылка;
    НоваяЗапись.Активность = Истина;
    НоваяЗапись.СчетДт = ПланыСчетов.Хозрасчетный.РасчетыСПрочимиПоставщикамиИПодрядчиками;
    НоваяЗапись.СубконтоДт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Контрагенты] = Справочники.Контрагенты.НайтиПоРеквизиту("ИНН","XXXXXXXXXXXX");
    НоваяЗапись.СубконтоДт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры] = Справочники.ДоговорыКонтрагентов.НайтиПоКоду("XXXXXX");
    НоваяЗапись.СубконтоДт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.ДокументыРасчетовСКонтрагентами] = Реализация;
    НоваяЗапись.СчетКт = ПланыСчетов.Хозрасчетный.РасчетыСПокупателями;
    НоваяЗапись.СубконтоКт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Контрагенты] = Реализация.Контрагент;
    НоваяЗапись.СубконтоКт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры] = Реализация.ДоговорКонтрагента;
    НоваяЗапись.СубконтоКт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.ДокументыРасчетовСКонтрагентами] = Реализация;
    НоваяЗапись.Сумма = Данные.СуммаДокумента;
    НаборЗаписей.Записать();
    УстановитьПривилегированныйРежим(Ложь);
    КонецЕсли;
```

Документ создается проводки есть но когда формирую оборотно-сальдовую ведомость по счету то этой операции нет. когда провожу документ интерактивно появляется. подскажите где собака закопана)) спасибо
---
я немного по другому делал, сразу добавлял движения к объекту.

```1c
    мОперация.Ответственный = ТекущийПользователь;
    мОперация.Комментарий = Комментарий;
    мОперация.Организация = Организация;
    мОперация.Дата = ЭлементОле.Дата;
    мОперация.СуммаОперации = ЭлементОле.Себестоимость;
    мОперация.Содержание="Списание запчастей";
    мОперация.Записать();

    //добавим проводки
    НовДв = мОперация.Движения.Хозрасчетный.Добавить();
    НовДв.Период = мОперация.Дата;
    НовДв.Регистратор = мОперация.Ссылка;
    НовДв.Организация = Организация;

    НовДв.СчетКт = ПланыСчетов.Хозрасчетный.НайтиПоКоду("10.05");

    НовДв.СубконтоКт.Вставить(НовДв.СчетКт.ВидыСубконто.Получить(2).ВидСубконто, мСклад);

    НовДв.Сумма = Окр(ЭлементОле.Себестоимость,2);
    НовДв.КоличествоКт = ЭлементОле.Количество;

    НовДв.СчетДт = ПланыСчетов.Хозрасчетный.НайтиПоКоду("20.01");
    НовДв.ПодразделениеДт = мПодразд;
    НовДв.СубконтоДт.Вставить(НовДв.СчетДт.ВидыСубконто.Получить(0).ВидСубконто, НГ);
    НовДв.СубконтоДт.Вставить(НовДв.СчетДт.ВидыСубконто.Получить(1).ВидСубконто, мСтатьяРасходов);

    НовДв.СубконтоКт.Вставить(НовДв.СчетКт.ВидыСубконто.Получить(0).ВидСубконто, мммМатериалы);
    мОперация.Движения.Хозрасчетный.Записывать = Истина;
    мОперация.Записать();
```
---
Если для 8.3 то
не

```1c
    ДокументОбъект.Проведен = Истина;
```

а

```1c
    ДокументОбъект.Движения.Хозрасчетный.Записывать = Истина;
```
---
я немного по другому делал, сразу добавлял движения к объекту.

```1c
    мОперация.Ответственный = ТекущийПользователь;
    мОперация.Комментарий = Комментарий;
    мОперация.Организация = Организация;
    мОперация.Дата = ЭлементОле.Дата;
    мОперация.СуммаОперации = ЭлементОле.Себестоимость;
    мОперация.Содержание="Списание запчастей";
    мОперация.Записать();

    //добавим проводки
    НовДв = мОперация.Движения.Хозрасчетный.Добавить();
    НовДв.Период = мОперация.Дата;
    НовДв.Регистратор = мОперация.Ссылка;
    НовДв.Организация = Организация;

    НовДв.СчетКт = ПланыСчетов.Хозрасчетный.НайтиПоКоду("10.05");

    НовДв.СубконтоКт.Вставить(НовДв.СчетКт.ВидыСубконто.Получить(2).ВидСубконто, мСклад);

    НовДв.Сумма = Окр(ЭлементОле.Себестоимость,2);
    НовДв.КоличествоКт = ЭлементОле.Количество;

    НовДв.СчетДт = ПланыСчетов.Хозрасчетный.НайтиПоКоду("20.01");
    НовДв.ПодразделениеДт = мПодразд;
    НовДв.СубконтоДт.Вставить(НовДв.СчетДт.ВидыСубконто.Получить(0).ВидСубконто, НГ);
    НовДв.СубконтоДт.Вставить(НовДв.СчетДт.ВидыСубконто.Получить(1).ВидСубконто, мСтатьяРасходов);

    НовДв.СубконтоКт.Вставить(НовДв.СчетКт.ВидыСубконто.Получить(0).ВидСубконто, мммМатериалы);
    мОперация.Движения.Хозрасчетный.Записывать = Истина;
    мОперация.Записать();
```
---
Попробуй посмотреть в консоли запросов движения от операции до интерактивного проведения и после. Может в каком-то поле есть разница.
---
Если для 8.3 то
не

```1c
    ДокументОбъект.Проведен = Истина;
```

а

```1c
    ДокументОбъект.Движения.Хозрасчетный.Записывать = Истина;
```
---
поправил код заработало всем спасибо

```1c
    ДокументОбъект = Документы.ОперацияБух.СоздатьДокумент();
    ДокументОбъект.Дата = ДатаПоступлениеНаРасчетныйСчет;
    ДокументОбъект.Организация = Данные.ОрганизацияМорозовИП;
    ДокументОбъект.СуммаОперации = Данные.СуммаДокумента;
    ДокументОбъект.Ответственный = ПараметрыСеанса.ТекущийПользователь;
    ДокументОбъект.Содержание = "УК "Равновесие" " + Реализация.Контрагент.Наименование;
    ДокументОбъект.Комментарий = "операция создана обработкой";
    ДокументОбъект.Проведен = Истина;
    Движения = ДокументОбъект.Движения.Хозрасчетный;
    Движение = Движения.Добавить();
    Движение.Активность = Истина;
    Движение.СчетДт = ПланыСчетов.Хозрасчетный.РасчетыСПрочимиПоставщикамиИПодрядчиками;
    Движение.СубконтоДт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Контрагенты] = Справочники.Контрагенты.НайтиПоРеквизиту("ИНН","5256089864");
    Движение.СубконтоДт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры] = Справочники.ДоговорыКонтрагентов.НайтиПоКоду("00-005381");
    Движение.СубконтоДт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.ДокументыРасчетовСКонтрагентами] = Реализация;
    Движение.СчетКт = ПланыСчетов.Хозрасчетный.РасчетыСПокупателями;
    Движение.СубконтоКт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Контрагенты] = Реализация.Контрагент;
    Движение.СубконтоКт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры] = Реализация.ДоговорКонтрагента;
    Движение.СубконтоКт[ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.ДокументыРасчетовСКонтрагентами] = Реализация;
    Движение.Сумма = Данные.СуммаДокумента;
    Движения.Записывать = Истина;
    Попытка
    ДокументОбъект.Записать(РежимЗаписиДокумента.Запись);
    Исключение
    Отказ = Истина;
    Сообщить(ОписаниеОшибки());
    КонецПопытки;
```
---
Помогло, спасибо!
---
ЗНАТОКИ, подскажите как из экселя перенести кучу начислений в Документ Операция, в один документ Операция кучу записей
---
Вот может эта обработка вам поможет в решении вашего вопроса: https://infostart.ru/public/1400921/